


```
> restart
```

```
> with(LinearAlgebra):
  with(Physics):
  with(plots):
  with(plottools):
  with(DEtools):
  with(Typesetting):
  interface(typesetting=extended):
  interface(showassumed=0):
```

```
> x_m:=<x(t)+l*sin(theta(t)), -l*cos(theta(t))>:
  print(`Mass m position vector`=x_m);

  printf("\n");

  v_m:=diff(x_m,t):
  print(`Mass m velocity vector`= v_m);
```

$$\textit{Mass m position vector} = \begin{bmatrix} x(t) + l\sin(\theta(t)) \\ -l\cos(\theta(t)) \end{bmatrix}$$

$$\textit{Mass m velocity vector} = \begin{bmatrix} \dot{x}(t) + l\,\dot{\theta}(t)\cos(\theta(t)) \\ l\,\dot{\theta}(t)\sin(\theta(t)) \end{bmatrix}$$

```
> T_m:=1/2*m*(v_m(1)^2+v_m(2)^2):
  V_m:=-m*g*l*cos(theta(t)):

  printf("\n");

  print(`Kinetic energy of mass m:`);
  print('T__m=1/2*m*(v[x]^2+v[y]^2)');
  T:=simplify(T_m):
  print('T__m'=T_m);

  printf("\n");

  print(`Potential energy of mass m:`);
  print('V__m'=V_m);

  printf("\n");
```

Kinetic energy of mass m:

$$T_m = \frac{1}{2}\, m \left(v_x^2 + v_y^2\right)$$

$$T_m = \frac{m\left(\left(\dot{x}(t) + l\,\dot{\theta}(t)\cos(\theta(t))\right)^2 + l^2\,\dot{\theta}(t)^2\sin(\theta(t))^2\right)}{2}$$

Potential energy of mass m:

$$V_m = -m\,g\,l\cos(\theta(t))$$

```
> T_M:=1/2*M*((diff(x(t),t))^2):
  V_M:=0:

  printf("\n");

  print(`Kinetic energy of mass M:`);
  print('T__M'=T_M);

  printf("\n");

  print(`Potential energy of mass M:`);
  print('V__M'=V_M);
```

Kinetic energy of mass M:

$$T_M = \frac{M\,\dot{x}(t)^2}{2}$$

Potential energy of mass M:

$$V_M = 0$$

```
> T:=T_m+T_M:
  V:=V_m+V_M:
  L:=T-V:
  print('L'=L);

  eq1:=simplify(<diff(L,x(t)), diff(L,theta(t))>):
  eq2:=simplify(<diff(L,diff(x(t),t)), diff(L,diff(theta(t),t))>):
  eq3:=simplify(<diff(eq2[1],t), diff(eq2[2],t)>):

  print('diff(L,x)'=eq1);
  print('diff(L,diff(x,t))'=eq2);
  print('diff(diff(L,diff(x,t)),t)'=simplify(eq3));

  printf("\n");
  print(`Thus,`);
  print('diff(diff(L,diff(x,t)),t)=diff(L,x)');
  eq4:=simplify({(eq3[1]-eq1[1]=0), (eq3[2]-eq1[2]=0)/(m*l)}):
  print(<eq4[1],eq4[2]>);
  eq5:=simplify(solve(eq4,{diff(x(t),t,t), diff(theta(t),t,t)})):
  print(<eq5[1],eq5[2]>);
```

$$L = \frac{m\left(\left(\dot{x}(t) + l\,\dot{\theta}(t)\cos(\theta(t))\right)^2 + l^2\,\dot{\theta}(t)^2\sin(\theta(t))^2\right)}{2} + \frac{M\,\dot{x}(t)^2}{2} + m\,g\,l\cos(\theta(t))$$

$$\frac{\partial}{\partial x} L = \begin{bmatrix} 0 \\ -m\, l \sin(\theta(t)) \left(\dot{\theta}(t)\, \dot{x}(t) + g\right) \end{bmatrix}$$

$$\frac{\partial}{\partial \frac{\partial}{\partial t} x} L = \begin{bmatrix} (M+m)\, \dot{x}(t) + \dot{\theta}(t) \cos(\theta(t))\, l\, m \\ m\, l \left(\cos(\theta(t))\, \dot{x}(t) + \dot{\theta}(t)\, l\right) \end{bmatrix}$$

$$\frac{\partial^2}{\partial \frac{\partial}{\partial t} x \partial t} L = \begin{bmatrix} (M+m)\, \ddot{x}(t) - m\, l \left(\dot{\theta}(t)^2 \sin(\theta(t)) - \ddot{\theta}(t) \cos(\theta(t))\right) \\ m\, l \left(-\dot{\theta}(t) \sin(\theta(t))\, \dot{x}(t) + \cos(\theta(t))\, \ddot{x}(t) + \ddot{\theta}(t)\, l\right) \end{bmatrix}$$

Thus,

$$\frac{\partial^2}{\partial \frac{\partial}{\partial t} x \partial t} L = \frac{\partial}{\partial x} L$$

$$\begin{bmatrix} (M+m)\, \ddot{x}(t) - m\, l \left(\dot{\theta}(t)^2 \sin(\theta(t)) - \ddot{\theta}(t) \cos(\theta(t))\right) = 0 \\ \ddot{\theta}(t)\, l + \cos(\theta(t))\, \ddot{x}(t) + \sin(\theta(t))\, g = 0 \end{bmatrix}$$

$$\begin{bmatrix} \ddot{\theta}(t) = -\dfrac{\sin(\theta(t)) \left(\dot{\theta}(t)^2 \cos(\theta(t))\, l\, m + g\, (M+m)\right)}{l \left(-\cos(\theta(t))^2 m + M + m\right)} \\ \ddot{x}(t) = \dfrac{\sin(\theta(t))\, m \left(\dot{\theta}(t)^2 l + \cos(\theta(t))\, g\right)}{-\cos(\theta(t))^2 m + M + m} \end{bmatrix}$$

We can not use the last system equations because they have singularity. Thus, we will use the previous system equations.

Let us take $\theta(t) = \omega t$, whic means, the small mass is rotating at a constant speed of ω . Thus, we can safely take only the second equation and ignore the first equation.

```
> eq6:=theta(t)=omega*t:
  eq7:=diff(eq6,t):
  eq8:=subs({eq6,eq7},eq5[2]):

  print(`Given:`):
  print(eq6):
  print(`Thus:`):
  print(eq8):
  print(`This is the final equation that we need to solve.`)
```

Given:

$$\theta(t) = \omega\, t$$

Thus:

$$\ddot{x}(t) = \frac{\sin(\omega\, t)\, m \left(\omega^2 l + \cos(\omega\, t)\, g\right)}{-\cos(\omega\, t)^2 m + M + m}$$

This is the final equation that we need to solve.

```
> OMEGA := 4:
GRAVITY := 9.8:
LENGTH := 0.05:
M:=10:
m:=1:
```

```
> ode:=eval(eq8, {l=LENGTH,g=GRAVITY,omega=OMEGA}):
sol:=dsolve({ode, x(0)=0, D(x)(0)=0}, x(t), numeric, method=rkf45, output=
listprocedure):
x_array:=eval(x(t),sol):

big_ball := proc(x,y) plots[pointplot]([[x,y]],color=blue,symbol=solidcircle,
symbolsize=200): end proc:
small_ball := proc(x,y) plots[pointplot]([[x,y]],color=green,symbol=solidcircle,
symbolsize=30): end proc:
```

```
> anim1:=animate(big_ball, [x_array(t),0], t=0..5,scaling=constrained):
anim2:=animate(small_ball, [LENGTH*sin(OMEGA*t)+x_array(t), -LENGTH*cos(OMEGA*t)], t=0.
.5,scaling=constrained):
anim3:=animate(plot,[[x_array(t),LENGTH*sin(OMEGA*t)+x_array(t)],[0, -LENGTH*cos(OMEGA*
t)],color=red], t=0..5, scaling=constrained):


h:=display(anim1,anim2,anim3);
```